
FW663 -- Laboratory Exercise

Basic Sampling

Background

This exercise will illustrate some fundamentals of sampling and data analysis theory to
supplement the statistical review during the first 3 lectures in class.  Further exercises will make
use of this exercise, so work carefully.  You are free to consult classmates on this exercise, but
when drawing your sample data, work independently.

First consider the physical model we will be sampling.  Buy 2 rolls of 50 pennies and
spread them out on a flat surface or table such that 30 are tails and 70 are heads.  We will use this
to mimic a small deer population where, at the end of a time period, 30 "deer" are dead (i.e., tails)
and 70 "deer" remain alive (i.e., heads).  Obviously, the population numbered 100 at the
beginning of the time period (unless the bank ripped you off!).

In this case we know some of the population parameters exactly (i.e., census results show
N = 100 and the parameter S = survival is obviously S = 70/100 = 0.70).  We now wish to
consider sampling from this population in an effort to estimate the parameter S from sample data.

First, it is important to carefully consider what we are doing and state some definitions. 
The population of interest is the "deer" population in a well-defined area (the table surface).  In
many cases the population size N is known, as it is here.  In other cases it is not known and
sometimes it is this parameter that we are trying to estimate.  The parameter S is defined as

S = finite survival rate (probability) of the population over
      a certain time period (e.g., 1 year).  

Quantitatively, S is defined as

S = no. alive at end of year / size of population at start of year

or S = A/N, where A = no. alive at end of year.

An obvious estimator of S (denoted with a "hat" to distinguish it from the parameter) is

 = a/nŜ

where a = no. of deer found alive in the sample at the end of the
    year (obtained by counting those in the sample),

n = total no. of all deer in the sample (no. dead plus no.
    alive) at the end of the year (again, obtained by counting
    all deer in the sample).

Sampling

In this exercise, you are to conduct 50 independent sample surveys to estimate the
parameter S.  Denote these estimates as .  Arrange your sampling procedure suchŜ1, Ŝ2, ..., Ŝ50
that n is about:

25 if your birthday is an odd number, and
70 if your birthday is an even number.

Thoroughly mix the dead and living deer before each sample of size n is drawn, being careful not
to flip any "deer" during the reshuffling.

Presumably you would rather not have to undergo the boredom of actually drawing 50
samples of 70 pennies from a table.  Thus for those of you feeling comfortable with the
microcomputers, the following SAS program is provided to do your penny counting.  However,
be sure to understand the physical model described above so that you understand what the
program is doing for you.  A copy of the program is stored on the network at the microlab as
J:\CLASSES\FW663\EXERCISE.02\PENNY.SAS.

data penny;
   samsiz=25;                 /* Enter your sample size (n) here */
                              /* 20-30 or 60-70                  */
   popsiz=100;                /* Population size is N=100        */
   array pop{100} $ 1;        /* Array pop holds the population  */
   seed=0;  /* seed set to zero initiates random number seed from clock */
   S=0.7;                     /* Population survival rate */
   do i=1 to S*popsiz;        /*  \                                */
      pop{i}="H";             /*  |                                */
      end;                    /* do loops to create the population */
   do i=S*popsiz+1 to popsiz; /* of pennies                        */
      pop{i}="T";             /*  |                                */
      end;                    /*  /                                */
   put "Population size (N) = " popsiz;
   * Loop to perform 500 simulations of the experiment ;
   do samnum=1 to 500;
      * Loop to shuffle the pennies in the population;
      * By shuffling them within the population, the sample;
      * can consist of the first "samsiz" of the population;
      do i=2 to popsiz;
         /* Pick penny to swap from list in [i,popsiz] with position i-1 */
         ii=int(ranuni(seed)*(popsiz-(i-1)))+i;
         /* Perform swap with these 3 statements */
         poptmp=pop{i-1};
         pop{i-1}=pop{ii};
         pop{ii}=poptmp;
         end;
      * List sample on screen in LOG window for first 10 samples;
      if samnum <= 10 then do;
         put "Sample number " samnum " is:";
         do i=1 to samsiz;
           put pop{i}=;
           end;
      end;
      * Cumulate the number of heads and tails in the sample;
      nheads=0; ntails=0;
      do i=1 to samsiz;
         if pop{i} = "H" then nheads=nheads+1;
         else ntails=ntails+1;
         end;
      if samnum <= 10 then do;

         put "Number of heads (alive)  = " nheads;
         put "Number of tails (deaths) = " ntails;
      end;
      Shat=nheads/samsiz;
      * Put results of this simulation into SAS file for later analysis;
      keep popsiz samsiz samnum S Shat;
      label S='Population survival rate'
            Shat='Estimate of survival rate'
            popsiz='Population size (N)'
            samsiz='Sample size (n)'
            samnum='Sample number or replicate';
      output;
      end; /* of samnum */
/* Output for 31 January assignment */
proc print; var samsiz samnum Shat;
proc chart; vbar Shat;
proc means maxdec=5 mean std stderr var min max; var Shat;

data penny2;
   set penny;
   label fvarShat='Finite population correction var(Shat)'
         fseShat='Finite population correction se(Shat)'
         varShat='Uncorrected var(Shat)'
         seShat='Uncorrected se(Shat)'
         uci='Upper 95% CI value'
         lci='Lower 95% CI value'
         cover='Coverage, 1=yes, 0=no';
   fvarShat=((popsiz - samsiz)/popsiz) * (Shat*(1 - Shat)/samsiz);
   fseShat=sqrt(fvarShat);
   varShat=Shat*(1 - Shat)/samsiz;
   seShat=sqrt(varShat);
   uci=Shat + tinv(0.975,samsiz-1)*fseShat;
   lci=Shat - tinv(0.975,samsiz-1)*fseShat;
   if uci > S & lci < S then cover=1;
   else cover=0;
proc means maxdec=5; var Shat fvarShat fseShat varShat seShat cover;
run;

Data Analysis

Prepare answers to the following questions to be discussed on Monday, 24 January, or
later as necessary.

1. The estimates Ŝi

2. The sample size you used (i.e., 25 or 70).

3. A simple histogram either on 10 lines/inch graph paper or constructed with SAS

4. Compute the mean of the estimates

Ave( ) = average( )Ŝ Ŝ1, Ŝ2, ..., Ŝ500

5. Bias cannot be assessed well with only 500 repetitions; however, you can get an idea
of the bias in the estimator by taking Ave( ) as an approximation to the expectedŜ
value of the estimator, E( ).  Then bias (b) can be computed approximately asŜ

b . Ave( ) - SŜ

Compute this quantity.  You may want to do more repetitions to completely check the
bias of the estimator.

6. Compute estimates of the sampling variance of the estimator  for each of yourŜ
surveys of dead deer.  The appropriate estimator is

ˆvar(Ŝ) '
N & n

N
Ŝ(1 & Ŝ)

n

Compute the standard error.  SAS would be appropriate for this, using programming
language commands.  Note that the term [(N - n)/N] is the finite population correction
(fpc).

7. Also, you may assume you do not know the population size, i.e., that the population is
large (N >> 0), so also compute the estimate of the sampling variance of the estimator
for each sample as

ˆvar(Ŝ) '
Ŝ(1 & Ŝ)

n

Compute the standard error, i.e.,  .ŜE ' vâr(Ŝ)

8. In all of the above, compute the 95% confidence interval in the usual (but certainly not
the only) way 

95% C. I. = Ŝ ± z vâr(Ŝ)
or

95% C. I. = Ave(Ŝ) ± z vâr(Ŝ)

where z = 1.96.  For n < 30, a t statistic would be appropriate instead of a z.

9. Compute the achieved coverage for the uncorrected and corrected estimates of ˆvar(Ŝ)
based on your estimates.

10. State your conclusions about coverage and compare the results with and without the
use of the fpc.

11. Based on the histogram you prepared, can you conclude the sampling distribution of
the estimator is normal? Symmetric? Skewed?

12. If you used the estimates of Ave( ) from everyone in the class, could you compute aŜ
very good approximation to E( )? Why? Why not?Ŝ

13. Is the estimator  in closed form?  Why?  Why not?Ŝ

14. Be certain you understand and can carefully define the following terms:  population,
parameter, estimator, estimate, accuracy, bias, precision, sample, survey, census,
expected value, statistic, sampling distribution of an estimator, sampling variance,
standard error, confidence interval, coverage, and coefficient of variation.

Work carefully, present your results clearly, make yourself a copy of the results and histogram
as they will be used again.  Remember, you can use the Microcomputer Laboratory during
daytime hours when a class is not in session, and during the evening.  SAS can perform all of
the calculations needed for questions 1-15.  However, you may feel more comfortable in doing
the calculations by hand or using a different computer program.

Helpful Literature

Pages 14-44 of White et al. 1982, LA-8787-NERP.


